

## Appendix A

### A.1 List of the Useful SPARQL Queries Overcoming the Required Objectives and Covering the Necessary Criteria

#### Query 1: Retrieve the algorithm of a given ML model

```
SELECT ?algo_name
WHERE
{
  ?ml rdf:type mlm:MachineLearningModel .
  ?ml mlm:hasAlgorithm ?algo .
  ?algo mlm:AlgorithmName ?algo_name .
  ?ml mlm:mlmID "1"^^xsd:string
}
```

#### Query 2: Describe the training data set of a given ML model

```
SELECT ?dataitem ?value ?featname
WHERE
{
  ?ml rdf:type mlm:MachineLearningModel .
  ?ml mlm:mlmID "1"^^xsd:string
  {
    ?ml mlm:hasTrainingDataSet ?tds .
    ?tds mlm:isComposedOf ?dataitem .
    ?dataitem mlm:dataitemValue ?value
  }
  UNION
  {
    ?ml mlm:hasTrainingDataSet ?tds .
    ?tds mlm:ReliesOn ?feature .
    ?feature mlm:featureName ?featname
  }
}
```

### Query 3: Describe the testing data set of a given ML model

```
SELECT ?dataitem ?value ?featname
WHERE
{
  ?ml rdf:type mlm:MachineLearningModel .
  ?ml mlm:mlmID "1"^^xsd:string
  {
    ?ml mlm:hasTestingDataSet ?tds .
    ?tds mlm:isComposedOf ?dataitem .
    ?dataitem mlm:dataitemValue ?value
  }
  UNION
  {?ml mlm:hasTestingDataSet ?tds .
  ?tds mlm:ReliesOn ?feature .
  ?feature mlm:featureName ?featname}
}
```

### Query 4: Retrieve the performance of a given ML model (the scores and the metrics used to calculate the evaluation applied to the testing data)

```
SELECT ?metric ?scorevalue
WHERE
{
  ?ml rdf:type mlm:MachineLearningModel .
  ?ml mlm:mlmID "1"^^xsd:string .
  ?ml mlm:hasTestingDataSet ?tds .
  ?tds mlm:hasEvaluation ?eval .
  ?eval mlm:hasScore ?score .
  ?score mlm:isCalculatedBy ?metric .
  ?score mlm:scoreValue ?scorevalue
}
```

**Query 5: Retrieve the metadata of a given ML model with those related to its algorithm / its training data set / its testing data set and to the evaluations applied to the testing data)**

```

SELECT ?mlmdfeaturename ?mlmdvalueval ?algo ?algomdfeaturename
?algomdvalueval ?trds ?trdsmdfeaturename
?trdsmdvalueval ?tsds ?tsdsmdfeaturename ?tsdsmdvalueval
?eval ?evalmdfeaturename ?evalmdvalueval
WHERE
{
  ?ml rdf:type mlm:MachineLearningModel .
  ?ml mlm:mlmID "1"^^xsd:string
  {
    ?ml mlm:hasAlgorithm ?algo .
    ?algo mlm:hasAlgorithmMetaData ?algo_md .
    ?algo_md mlm:hasFeature ?algomdfeature .
    ?algomdfeature mlm:featureName ?algomdfeaturename .
    ?algo_md mlm:hasValue ?algomdvalue .
    ?algomdvalue mlm:valueVal ?algomdvalueval
  }
  UNION
  {
    ?ml mlm:hasTrainingDataSet ?trds .
    ?trds mlm:DSHasMetaData ?trds_md .
    ?trds_md mlm:hasFeature ?trdsmdfeature .
    ?trdsmdfeature mlm:featureName ?trdsmdfeaturename .
    ?trds_md mlm:hasValue ?trdsmdvalue .
    ?trdsmdvalue mlm:valueVal ?trdsmdvalueval
  }
  UNION
  {
    ?ml mlm:hasTestingDataSet ?tsds .
    ?tsds mlm:DSHasMetaData ?tsds_md .
    ?tsds_md mlm:hasFeature ?tsdsmdfeature .
    ?tsdsmdfeature mlm:featureName ?tsdsmdfeaturename .
    ?tsds_md mlm:hasValue ?tsdsmdvalue .
    ?tsdsmdvalue mlm:valueVal ?tsdsmdvalueval .
    ?tsds mlm:hasEvaluation ?eval .
    ?eval mlm:hasEvaluationMetaData ?eval_md .
    ?eval_md mlm:hasFeature ?evalmdfeature .
    ?evalmdfeature mlm:featureName ?evalmdfeaturename .
    ?eval_md mlm:hasValue ?evalmdvalue .
    ?evalmdvalue mlm:valueVal ?evalmdvalueval
  }
  UNION
  {
    ?ml mlm:hasMetaData ?ml_md .
    ?ml_md mlm:hasFeature ?mlmdfeature .
    ?mlmdfeature mlm:featureName ?mlmdfeaturename .
    ?ml_md mlm:hasValue ?mlmdvalue .
    ?mlmdvalue mlm:valueVal ?mlmdvalueval
  }
}

```

**Query 6: Find the application domain of each ML model and give a clear description of this domain**

```
SELECT ?ml ?domain_name (GROUP_CONCAT(?featurename;SEPARATOR=",") AS
    ↪ ?features)
WHERE
{
    ?ml rdf:type mlm:MachineLearningModel .
    ?ml mlm:hasApplicationDomain ?appdomain .
    ?appdomain mlm:domainName ?domain_name .
    ?appdomain mlm:isDescribedBy ?topic .
    ?topic mlm:hasTopicFeature ?feature .
    ?feature mlm:featureName ?featurename
}
GROUP BY ?ml ?domain_name
```

**Query 7: Describe the training data context of a given ML model**

```
SELECT ?featurename ?operatorvalue ?featurevalue
WHERE
{
    ?ml rdf:type mlm:MachineLearningModel .
    ?ml mlm:mlmID "1"^^xsd:string .
    ?ml mlm:hasTrainingDataSet ?tds .
    ?tds mlm:hasContext ?cont .
    ?cont mlm:hasConstraint ?constr .
    ?constr mlm:hasSourceOperand ?feature .
    ?feature mlm:featureName ?featurename.
    ?constr mlm:hasOperator ?op .
    ?op mlm:operatorValue ?operatorvalue .
    ?constr mlm:hasTargetOperand ?value .
    ?value mlm:valueVal ?featurevalue
}
```